
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2015-03_D15wk25’.

3. Corrections to generic CA Cell Initialization function
Change 1 – Correction to function ‘f_EUTRA_CA_SetDlUlBandwidth’
	Function name
	f_EUTRA_CA_SetDlUlBandwidth ()

	Reason for change
	1.   The loop indexing proposed in earlier R5s150438 is wrong and leads to a cell initialization problem in case of CA. 
2.    Inappropriate usage of the ispresent TTCN function. It should not be used to check the presence of a list element, rather the length of the list should be checked. The problem is that by using the ispresent, the TTCN compiler extends the list by an additional element which is set to omit and this leads to a wrong list size including a redundant element at the end.

	Summary of change
	1.   Amended the loop indexing.
2.   Already covered by R5s150663.   =>  no need for action

	TTCN module
	\Common\EUTRA\EUTRA_CellInfoInit_CA.ttcn

	MCC160 Comment
	1. Accepted
2. Not needed, see R5s150663


Before change

...
  /*
   * @desc      Update the DL/UL bandwidth and all dependant bandwidth parameters in sysinfo and physical layer parameters
   * @param     p_CellIdList
   * @param     p_DL_ChBandwidth
   * @param     p_UL_ChBandwidth
   * @param     p_SysinfoCombination
   * @status    APPROVED (LTE_A)
   */
   function f_EUTRA_CA_SetDlUlBandwidth(EUTRA_CellIdList_Type   p_CellIdList,
                                        Dl_Bandwidth_Type p_DL_ChBandwidth,
                                        Ul_Bandwidth_Type p_UL_ChBandwidth,
                                        Combination_Type p_SysinfoCombination) runs on EUTRA_PTC
  { //@sic R5s150438 sic@
    var integer i;
    var integer v_CellListLength := lengthof (p_CellIdList);
    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(p_DL_ChBandwidth, p_UL_ChBandwidth);
    var template (value) EUTRA_CellInfo_Type v_CellInfo;
    var template (value) SoundingRS_UL_ConfigCommon v_SoundingRS_UL_ConfigCommon;
    var template (value) PRACH_ConfigSIB v_PRACH_ConfigSIB;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var RootSequenceIndex_Type v_RootSequenceIndex;
    for (i:=0; i<v_CellListLength - 1; i:=i+1) {  //It needs to start from the first element, otherwise the first cell is skipped
      v_CellInfo := f_EUTRA_CellInfo_Get(p_CellIdList[i]);
      v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(p_CellIdList[i]);
      v_RootSequenceIndex := f_EUTRA_InitialisePRACH_RootSequenceIndex(p_CellIdList[i], v_EUTRA_FDD_TDD_Mode, LTE_AllCellsOnSamePLMN);
      //Set DL/UL bandwidth
      v_CellInfo.Frequency.DL_ChBandwidth := p_DL_ChBandwidth;
      v_CellInfo.Frequency.UL_ChBandwidth := p_UL_ChBandwidth;
      //SIB3_periodicity
      v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[1].si_Periodicity := v_ChannelBandwidthDependency.SIB3_periodicity;
      select (p_SysinfoCombination) {
        case(c3) {
          //SIB5_periodicity
          if ( ispresent(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2])) {  //The given cell may not have this SIB initialised and periodicity alone cannot be set otherwise it will cause instability
            v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].si_Periodicity := v_ChannelBandwidthDependency.SIB5_periodicity;
          }
        }
        case(c4) {
          //SIB6_periodicity
          v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].si_Periodicity := v_ChannelBandwidthDependency.SIB6_periodicity;
        }
        case(c9) {
          //SIB5_periodicity/SIB6_periodicity
          v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].si_Periodicity := v_ChannelBandwidthDependency.SIB6_periodicity;
        }
        case else { FatalError (__FILE__, __LINE__, "invalid sysinfo combination");}
      }
...
After change

...
  /*
   * @desc      Update the DL/UL bandwidth and all dependant bandwidth parameters in sysinfo and physical layer parameters
   * @param     p_CellIdList
   * @param     p_DL_ChBandwidth
   * @param     p_UL_ChBandwidth
   * @param     p_SysinfoCombination
   * @status    APPROVED (LTE_A)
   */
   function f_EUTRA_CA_SetDlUlBandwidth(EUTRA_CellIdList_Type   p_CellIdList,
                                        Dl_Bandwidth_Type p_DL_ChBandwidth,
                                        Ul_Bandwidth_Type p_UL_ChBandwidth,
                                        Combination_Type p_SysinfoCombination) runs on EUTRA_PTC
  { //@sic R5s150438 sic@
    var integer i;
    var integer v_CellListLength := lengthof (p_CellIdList);
    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(p_DL_ChBandwidth, p_UL_ChBandwidth);
    var template (value) EUTRA_CellInfo_Type v_CellInfo;
    var template (value) SoundingRS_UL_ConfigCommon v_SoundingRS_UL_ConfigCommon;
    var template (value) PRACH_ConfigSIB v_PRACH_ConfigSIB;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var RootSequenceIndex_Type v_RootSequenceIndex;
    for (i:=0; i<v_CellListLength; i:=i+1) {
      v_CellInfo := f_EUTRA_CellInfo_Get(p_CellIdList[i]);
      v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(p_CellIdList[i]);
      v_RootSequenceIndex := f_EUTRA_InitialisePRACH_RootSequenceIndex(p_CellIdList[i], v_EUTRA_FDD_TDD_Mode, LTE_AllCellsOnSamePLMN);
      //Set DL/UL bandwidth
      v_CellInfo.Frequency.DL_ChBandwidth := p_DL_ChBandwidth;
      v_CellInfo.Frequency.UL_ChBandwidth := p_UL_ChBandwidth;
      //SIB3_periodicity
      v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[1].si_Periodicity := v_ChannelBandwidthDependency.SIB3_periodicity;
      select (p_SysinfoCombination) {
        case(c3) {
          //SIB5_periodicity
          if ( lengthof(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList) > 2 ) {
            v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].si_Periodicity := v_ChannelBandwidthDependency.SIB5_periodicity;
          }
        }
        case(c4) {
          //SIB6_periodicity
          v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].si_Periodicity := v_ChannelBandwidthDependency.SIB6_periodicity;
        }
        case(c9) {
          //SIB5_periodicity/SIB6_periodicity
          v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.schedulingInfoList[2].si_Periodicity := v_ChannelBandwidthDependency.SIB6_periodicity;
        }
        case else { FatalError (__FILE__, __LINE__, "invalid sysinfo combination");}
      }
... 
